
Example 13.7

Ordering Simulation Models

Background Information
n Home repair is a large hardware retail store for more 

hammers.

n The setup cost for placing an order is $500, 
independent of the size of the order. The unit cost per 
hammer is $20.

n Home Repair estimates that the cost of holding a 
hammer in inventory for 1 week is $3.

n The company defines its inventory position at the 
beginning of any week as the number of hammers in 
inventory plus any that have already been ordered 
but have not yet arrived.

Background Information --
continued
n The company’s ordering policy is an (s,S) policy, a 

common periodic review policy used by many 
companies.

n This policy, defined by two numbers s and S, where 
s<S, specifies that if the inventory position at the 
beginning of the week is at level x, where x is less 
than or equal to s, exactly enough hammers will be 
ordered to bring the inventory position up to S – that 
is, Home Repair will order S – s hammers.

13.1 | 13.2 | 13.3 | 13.4 | 13.5 | 13.6 |13.8

Background Information --
continued
n Otherwise, if the inventory position is greater than s, 

no order will be placed that week.

n If an order is placed, it will arrive after a lead time of 
1, 2, or 3 weeks with probabilities 0.7, 0.2, and 0.1.

n The weekly demand for hammers is uncertain, but it 
can be described by a normal distribution with mean 
300 and standard deviation 75.

n The company’s policy is to satisfy all demand in the 
week it occurs.

Background Information --
continued
n If weekly demand cannot be satisfied completely from 

on-hand inventory, then an emergency order will be 
placed at the end of the week for the shortage.

n This order will arrive virtually instantaneously, but at a 
step cost of $35 per hammer.

n It is currently the beginning of week 1, and the 
current inventory of hammers, including any that 
might just have arrived, is 600.

n There are no other orders on the way. Home Repair 
wants to simulate several (s,S) policies to see which 
does best in terms of total costs over the next 52 
weeks.

Solution
n We will use @Risk to simulate a 52-week period and 

keep track of the total costs for this period for each of 
several (s,S) policies.

n There is no way to optimize all possible (s,S) policies, 
but it is possible to test a number of “representative” 
policies and choose the best of these.

ORDERSIM.XLS
n The simulation model is shown on the next slides.

n This file contains the model.

Developing the Model
n It is mostly a matter of careful bookkeeping, as we 

describe in the following steps. 

– Inputs. Enter the inputs in the shaded ranges. These include 
the various costs, the parameters of the demand distribution, 
the current inventory situation, and possible combinations of 
s and S to test. Note that the values in cells B30 and B31 are 
0 because we have assumed that no orders are currently on 
the way. However, we develop the model so that it could 
respond to nonzero values in these cells. They would 
correspond to orders placed before week 1 but not due in 
until after week 1.

Developing the Model --
continued

– Ordering policy. As usual, we set up the model with a 
RISKSIMTABLE function so that we can test all of the 
selected ordering policies simultaneously. To do this, enter 
the formula =RISKSIMTABLE(E29:E36) in cell B34. Then 
enter the formulas =VLOOKUP(B34,PolicyTable,2) and 
=VLOOKUP(B34,PolicyTable,3) in cells B34 and B35 to 
capture the values of s and S that will be used in the 
simulation.

Developing the Model --
continued

– Beginning inventory. Moving to the simulation model, our 
strategy is the same as in most multiperiod models. We fill in 
the logic for the first few weeks and then copy down. We 
begin with column B, which contains the beginning on-hand 
inventory, right after any order has arrived. For week 1, this 
is the initial 600 hammers, so enter the formula =InitInv in 
cell B44. For later weeks we have to sum the final inventory 
from the previous week and the amount due in, if any, from 
previous orders. To do this, enter enter the formulas 
=H44+Due2+SUMIF($F$44:F44,A45,$E$44:E44), 
=H45+Due3+SUMIF($F$44:F45,A46,$E$44:E45) and 
=H46+SUMIF($F$44:F46,A47,$E$44:E46) in cells B45-B47. 
This last formula is general, so copy it down to the other 
weeks.

Developing the Model --
continued

– Due in. In column C we record the amounts already ordered 
but not yet in, so that we can calculate the inventory position 
in column D. Do this by entering the formulas =Due2+Due3, 
=Due3+SUMIF($F$44:F44,”>”&A45,$E$44:E44) and 
=SUMIF($F$44:F45,”>”&A46,$E$44:E45) in cells C44-C46, 
and copy this latter formula down. The SUMIF function is 
used essentially as in the previous step, but now we want 
conditions (the middle argument) such as “>1”. TO do this in 
Excel, we must put the greater than sign in quotes, followed 
by an ampersand, and then a cell reference.

– Inventory position. The inventory position is the amount 
onhand plus the amount due in, so enter the formula 
=B44+C44 in cell D44 and copy it down.

Developing the Model --
continued

– Order. Following the logic of the (s, S) ordering policy, 
calculate the order quantity in cell E44 with the formula 
=IF(D44<=ReorderPt,OrderUpToQ-D44,0) and copy it 
down. Then to see when this order arrives, enter the formula 
=IF(E44>0,A44+RISKDISCRETE(LeadTimes,Probs),”NA”)
in cell F44 and copy it down.

– Demand. Generate random demands in column G by 
entering the formula 
=ROUND(RISKNORMAL(MeanDem,StdevDem),0) in cell 
G44 and copying it down.

Developing the Model --
continued

– Ending inventory and emergency orders. If customer 
demand is less than on-hand inventory, then ending 
inventory is the difference; otherwise it is 0. Therefore, enter
the formula =MAX(B44-G44,0) in cell H44 and copy it down. 
Similarly, there are emergency orders only if customer 
demand is greater than onhand inventory, so enter the 
formula =MAX(G44-B44,0) in cell I44 and copy it down.

– Weekly costs. The weekly costs are straightforward. 
Calculate them for week 1 in cells J44-M44 with the formulas 
=IF(E44>0,Fcost,0), =Vcost*E44, =Hcost*(B44+H44)/2 and 
=Ecost*I44 and then copy these down. 

Developing the Model --
continued

– Summary measures. Calculate the total cost of the various 
types in row 40 and designate them as @Risk output cells. 
For example, the formula in cell B40 is =RISKOUTPUT( 
)+SUM(J44:J95)

n It is important to look at the completed model before 
running @Risk. Press F9 a couple of times to get 
random numbers and check the logic.



9

13.1 | 13.2 | 13.3 | 13.4 | 13.5 | 13.6 |13.8

Using @Risk
n We use @Risk exactly as in Chapter 11 and 12. We 

set the number of iterations to 500 and the number of 
simulations to 8.

n After running @Risk and copying selected outputs 
back to Excel, we obtain the results on the next slide.

n The two shaded cells correspond to the smallest 
average total 52-week costs among all pairs of s and 
S.

n Home Repair might prefer the policy with s=500 and 
S=750.

Using @Risk -- continued

n It has the smallest average total cost, it has the 
smallest 5th percentile, it is essentially ties for the 
smallest median, and its 95th percentile is close to the 
smallest.

n Even with this ordering policy, however, there is still 
considerably variability – from about $364,000 for the 
best of the 500 iterations to about $454,000 for the 
worst.


